# Genetic groups model

**Animal model with genetic groups**

**Alternative method**

## Animal Model with Genetic Groups

$$y = Xb + Zu + ZQg + e$$

$$\begin{bmatrix} X'X & X'Z & X'ZQ \\ Z'X & Z'Z+\alpha A^{-1} & Z'ZQ \\ Q'Z'X & Q'Z'Z & Q'Z'ZQ \end{bmatrix} \begin{bmatrix} b \\ u \\ g \end{bmatrix} = \begin{bmatrix} X'y \\ Z'y \\ Q'Z'y \end{bmatrix}$$

## Data Description

Data file:

| ID | sex | wwt(kg) |
|---|---|---|
| 4 | M | 4.5 |
| 5 | F | 2.9 |
| 6 | F | 3.9 |
| 7 | M | 3.5 |
| 8 | M | 5.0 |

Assign unknown sires and dam with unrelated phantom parents.

Pedigree file:

| anim | s | d |
|---|---|---|
| 1 | 0 | 0 |
| 2 | 0 | 0 |
| 3 | 0 | 0 |
| 4 | 1 | 0 |
| 5 | 3 | 2 |
| 6 | 1 | 2 |
| 7 | 4 | 5 |
| 8 | 3 | 6 |

| an | s | d |
|---|---|---|
| 1 | p1 | p2 |
| 2 | p3 | p4 |
| 3 | p5 | p6 |
| 4 | 1 | p7 |
| 5 | 3 | 2 |
| 6 | 1 | 2 |
| 7 | 4 | 5 |
| 8 | 3 | 6 |

Let assign the simple grouping with the asumption that unknown sire and dam are from different groups

| | |
|---|---|
| p1,p3,p5 | in group1 |
| p2,p4,p6,p7 | in group2 |

Va = 20 Ve = 40

alpha = Ve/Va

alpha = (1-h2)/h2

## Start computing

```
PROC IML;
  X = {1 0,
       0 1,
       0 1,
       1 0,
       1 0};

  Z = {0 0 0 1 0 0 0 0,
       0 0 0 0 1 0 0 0,
       0 0 0 0 0 1 0 0,
       0 0 0 0 0 0 1 0,
       0 0 0 0 0 0 0 1};

  y = {4.5,2.9,3.9,3.5,5.0};
```

```
A = {1   0   0   .5    0   .5  .25 .25,
     0   1   0   0    .5   .5  .25 .25,
     0   0   1   0    .5    0  .25 .5,
    .5   0   0   1    0   .25 .5  .125,
     0   .5  .5  0    1   .25 .5  .375,
    .5   .5   0  .25  .25   1  .25 .5,
    .25 .25 .25 .5    .5   .25  1  .25,
    .25 .25 .5  .125 .375 .5  .25  1};
```

Relationship for all animals and unknown parents is required

**The following column are p1,p2,...p7,a1,a2,...a8**

```
A2 ={1    0    0    0    0    0    0  .5   0   0  .25   0   .25 .125 .125,
     0    1    0    0    0    0    0  .5   0   0  .25   0   .25 .125 .125,
     0    0    1    0    0    0    0   0  .5   0   0   .25  .25 .125 .125,
     0    0    0    1    0    0    0   0  .5   0   0   .25  .25 .125 .125,
     0    0    0    0    1    0    0   0   0  .5   0   .25   0  .125 .25 ,
     0    0    0    0    0    1    0   0   0  .5   0   .25   0  .125 .25 ,
     0    0    0    0    0    0    1   0   0   0  .5    0    0  .25   0  ,
    .5   .5    0    0    0    0    0   1   0   0  .5    0   .5  .25  .25 ,
     0    0   .5   .5    0    0    0   0   1   0   0   .5   .5  .25  .25 ,
     0    0    0    0   .5   .5    0   0   0   1   0   .5    0  .25  .5  ,
    .25  .25   0    0    0    0   .5  .5   0   0   1    0   .25 .5   .125,
     0    0   .25  .25  .25  .25   0   0   .5  .5   0    1   .25 .5   .375,
    .25  .25  .25  .25   0    0    0  .5   .5   0  .25  .25   1  .25  .5  ,
    .125 .125 .125 .125 .125 .125 .25 .25 .25 .25 .5   .5   .25  1   .25 ,
    .125 .125 .125 .125 .25  .25   0  .25 .25 .5  .125 .375 .5  .25   1 };
```

- **Next step is find T**

  **where A2 = TDT`**

- **Since we can find L**

  **A2 = LL`**

  **L=T*sqrt(D)**

  **Therefore T = L*INV(sqrt(D))**

  **T = L if ignored inbreeding**

- **Therfore we can find T from Cholesky decomposition which make LL`=A2**

```
T = HALF(A2);
```

Next step is define G

G is genetic group coding for all animals and note that only unknown parents are coding

```
G = {1 0,
     0 1,
     1 0,
     0 1,
     1 0,
     0 1,
     0 1,
     0 0,
     0 0,
     0 0,
     0 0,
     0 0,
     0 0,
     0 0,
     0 0,
     0 0};
```

Q is defined from TG

Function HALF give upper triangular, so need transpose

Then select only part of true animal

```
T = HALF(A2);

Qinit  = T`*G;
Q      = Qinit[8:15,];
```

Note:

Q can be simply generated without define T by relating each animal to matrix G

## MME Setup

```
XPX    = X`*X;
XPZ    = X`*Z;
ZPZ    = Z`*Z;
ZPZ2   = Z`*Z+alpha#Ai;
XPZQ   = X`*Z*Q;
QPZPZQ = Q`*Z`*Z*Q;

lhs = (X`*X    || X`*Z          || X`*Z*Q    )//
      (Z`*X    || Z`*Z+alpha#Ai || Z`*Z*Q    )//
      (Q`*Z`*X || Q`*Z`*Z       || Q`*Z`*Z*Q);

rhs = X`*y // Z`*y // Q`*Z`*y;
```

## MME Setup

```
lhs[1:12,11]=0;                Set genetic group1 to zero
lhs[11,1:12]=0;
rhs[11,]=0;
sol = GINV(lhs)*rhs;
```

## Compute accuracy

```
Di  = vecdiag(GINV(lhs));
PEV = Di#Ve;
I   = J(12,1,1);
Acc = J(12,1,.);
Acc[3:10,]  = SQRT(I[3:10,]-Di[3:10,]#alpha);
```

## Construct BV based on Groups

```
BV  = sol[3:10,];                Select BV for true animals
gr  = sol[11:12,];               Select group estimates
BVG = BV+Q*gr;                   Compute BV based on groups
BVG = J(2,1,.)//BVG//J(2,1,.);   Add missing for fix and gr
```

# Output

| SOL | | BVG | DI | PEV | ACC |
|---|---|---|---|---|---|
| b1 | 5.458 | . | 18.309 | 732.359 | . |
| b2 | 4.313 | . | 12.908 | 516.311 | . |
| u1 | 0.117 | -0.767 | 0.476 | 19.052 | 0.218 |
| u2 | -0.039 | -0.923 | 0.498 | 19.917 | 0.064 |
| u3 | -0.079 | -0.963 | 0.477 | 19.083 | 0.214 |
| u4 | 0.059 | -1.268 | 0.494 | 19.763 | 0.109 |
| u5 | -0.215 | -1.099 | 0.440 | 17.616 | 0.345 |
| u6 | 0.157 | -0.728 | 0.448 | 17.932 | 0.322 |
| u7 | -0.233 | -1.338 | 0.446 | 17.834 | 0.329 |
| u8 | 0.116 | -0.768 | 0.486 | 19.459 | 0.165 |
| g1 | 0.000 | . | 0.000 | 0.000 | . |
| g2 | -1.769 | . | 45.834 | 1833.367 | . |

Animal Model with Genetic Groups
(Alternative Method)

$$y = Xb + Zu + ZQg + e$$

$$\begin{bmatrix} X'X & X'Z & 0 \\ Z'X & Z'Z+\alpha A_{11} & \alpha A_{12} \\ 0 & \alpha A_{21} & \alpha A_{22} \end{bmatrix} \begin{bmatrix} b \\ u \\ g \end{bmatrix} = \begin{bmatrix} X'y \\ Z'y \\ 0 \end{bmatrix}$$

$A_{11}=inv(A)$, $A_{12}=-A_{11}Q$, $A_{22}=Q'A_{11}Q$

# Data Description

Data file:

| ID | sex | wwt(kg) | an | s | d |
|---|---|---|---|---|---|
| 4 | M | 4.5 | 1 | p1 | p2 |
| 5 | F | 2.9 | 2 | p3 | p4 |
| 6 | F | 3.9 | 3 | p5 | p6 |
| 7 | M | 3.5 | 4 | 1 | p7 |
| 8 | M | 5.0 | 5 | 3 | 2 |
| | | | 6 | 1 | 2 |
| | | | 7 | 4 | 5 |
| | | | 8 | 3 | 6 |

Pedigree file:

| anim | s | d |
|---|---|---|
| 1 | 0 | 0 |
| 2 | 0 | 0 |
| 3 | 0 | 0 |
| 4 | 1 | 0 |
| 5 | 3 | 2 |
| 6 | 1 | 2 |
| 7 | 4 | 5 |
| 8 | 3 | 6 |

p1,p3,p5 in group1
p2,p4,p6,p7 in group2

```
Q  = {.5   .5,
      .5   .5,
      .5   .5,
      .25  .75,
      .5   .5,
      .5   .5,
      .375 .625,
      .5   .5};

Ai11   = Ai;
Ai12   = -Ai*Q;
Ai21   = Ai12`;
Ai22   = Q`*Ai*Q;
```

## MME Setup

```
XPX    = X`*X;
XPZ    = X`*Z;
ZPZ    = Z`*Z;
ZPZ2   = Z`*Z+alpha#Ai11;

 ZERO   = J(2,2,0);
```

Create zero matrix for lhs relate to groups
row=2 levels for fix, col= 2 levels for groups

```
lhs = (X`*X   || X`*Z             || ZERO      )//
      (Z`*X   || Z`*Z+alpha#Ai11 || alpha#Ai12)//
      (ZERO`  || alpha#Ai21      || alpha#Ai22);
```

Set genetic group I mean to zero

```
lhs[11,1:12]=0;
lhs[1:12,11]=0;
```

Add zero value for 2 groups for RHS

```
rhs = X`*y // Z`*y // J(2,1,0);
sol = GINV(lhs)*rhs;
```

## Alternative method *vs.* Usual method

| SOL | |
|---|---|
| b1 | 5.458 |
| b2 | 4.313 |
| u1 | -0.767 |
| u2 | -0.923 |
| u3 | -0.963 |
| u4 | -1.268 |
| u5 | -1.099 |
| u6 | -0.728 |
| u7 | -1.338 |
| u8 | -0.768 |
| g1 | 0.000 |
| g2 | -1.769 |

| SOL | | BVG |
|---|---|---|
| b1 | 5.458 | . |
| b2 | 4.313 | . |
| u1 | 0.117 | -0.767 |
| u2 | -0.039 | -0.923 |
| u3 | -0.079 | -0.963 |
| u4 | 0.059 | -1.268 |
| u5 | -0.215 | -1.099 |
| u6 | 0.157 | -0.728 |
| u7 | -0.233 | -1.338 |
| u8 | 0.116 | -0.768 |
| g1 | 0.000 | . |
| g2 | -1.769 | . |